



Expert Tutorial

Using SAS PROC TCALIS for multigroup structural equation modelling with mean structures

Fei Gu^{1*} and Wei Wu²

¹Department of Psychology and Research in Education, University of Kansas, USA

²Department of Psychology, University of Kansas, USA

Multigroup structural equation modelling (SEM) is a technique frequently used to evaluate measurement invariance in social and behavioural science research. Before version 9.2, SAS was incapable of handling multigroup SEM. However, this limitation is resolved in PROC TCALIS in SAS 9.2. For the purpose of illustration, this paper provides a step-by-step guide to programming the tests of measurement invariance and partial invariance using PROC TCALIS for multigroup SEM with mean structures. Fit indices and parameter estimates are validated, thus providing an alternative tool for researchers conducting both applied and simulated studies. Other new features (e.g., different types of modelling languages and estimation methods) and limitations (e.g., ordered-categorical SEM and multilevel SEM) of the TCALIS procedure are also briefly discussed.

1. Introduction

Multigroup structural equation modelling (SEM) is a technique frequently used to evaluate measurement invariance in social and behavioural science research. In the past decades, a variety of commercial software packages have been developed for SEM, including EQS (Bentler & Wu, 2002), *Mplus* (Muthén & Muthén, 1998–2007), LISREL (Jöreskog & Sörbom, 1996), Mx (Neale, Boker, Xie, & Maes, 2003), and AMOS (Arbuckle, 2003). Since version 8, SAS has also added a procedure into the SAS/STAT product to accommodate SEM (i.e., PROC CALIS). However, the CALIS procedure has one major limitation – its inability to handle multigroup SEM (Fan & Fan, 2005). Although some researchers have tried to ‘trick’ SAS into analysing multigroup models (provided that each group had the same sample size), this trick is not generalizable to situations where unequal sample sizes are encountered (Jones-Farmer, Pitts, & Rainer, 2008; Marcoulides

*Correspondence should be addressed to Fei Gu, Department of Psychology and Research in Education, University of Kansas, 1122 West Campus Road, Lawrence, KS 66045, USA (e-mail: fgu@ku.edu).

& Hershberger, 1997). In addition, using such a trick may give incorrect degrees of freedom. Thus, one must be cautious about using PROC CALIS for multigroup analyses.

This limitation means that, for some researchers, SAS is not the first choice when it comes to implementing multigroup invariance tests (e.g., Jones-Farmer *et al.*, 2008). According to Byrne (2004), most literature addressing multigroup invariance has used either LISREL or EQS. However, there might be times when convenience or necessity dictates a preference for SAS. For example, SAS offers quantitative researchers an extremely flexible environment for conducting various Monte Carlo simulation studies (Fan, Felsövälyi, Sivo, & Keenan, 2003). Data simulation and subsequent analyses of the simulated results can be easily implemented in SAS by using a wide variety of descriptive and/or advanced statistical procedures (e.g., PROC MEANS and PROC GLM). Because of the above-mentioned limitation, simulation studies using PROC CALIS are confined to single-group analyses (see Fan & Sivo, 2005; Yang & Green, 2010). Simulation studies involving multigroup SEM can become very laborious if one needs to simulate data in SAS, export the data to another software package, say LISREL, for multigroup analyses, and then import the output back into SAS for subsequent analyses (e.g., Fan & Sivo, 2009). Generally, a practice that involves data exchange among different software packages is inconvenient and time-consuming in simulation studies.

SAS 9.2 saw the introduction of an experimental procedure, PROC TCALIS. This is an updated version of PROC CALIS, incorporating changes and enhancements. According to the SAS documentation (SAS Institute, 2008), PROC TCALIS is not a simple functional enhancement of PROC CALIS. The basic computational architecture of PROC TCALIS is quite different from that of PROC CALIS. New features include, but are not limited to, new modelling languages, multigroup analysis, and improved mean structures analysis. With the TCALIS procedure available, the inconvenient data exchange between SAS and other SEM packages in simulation studies can be avoided.

Nevertheless, up to the time of writing, no journal paper or textbook has provided an example program to illustrate the TCALIS procedure for multigroup SEM. Therefore, the purpose of this paper is to provide a step-by-step tutorial on using SAS PROC TCALIS. Specifically, tests of invariance and partial invariance of mean and covariance structures between two groups are illustrated with an empirical example. We believe that researchers who conduct both applied and simulated studies can benefit from such an alternative tool in their future work.

2. Measurement invariance

SEM models are used to describe the relationships between manifest and/or latent variables. When a particular theoretical model is justified as a good enough approximation to the sample data for a homogeneous group, the research question whether the same model holds across heterogeneous groups may be of interest. Such groups may be defined by any categorical variable in practice (e.g., gender, race/ethnicity, socioeconomic status, etc.). Testing measurement invariance in the multigroup framework is becoming increasingly popular for answering such related questions. Technically, measurement invariance can be tested at different levels. Detailed discussion of measurement invariance can be found in the literature elsewhere (e.g., Bollen, 1989; Byrne, Shavelson, & Muthén, 1989; Cheung & Rensvold, 2002; Little, 1997; Meredith, 1993; Vandenberg & Lance, 2000), and a brief summary is provided below.

The first level of invariance is that two or more groups share the same model structure but different parameter estimates across groups (*configural invariance*). If the configural model adequately fits the data, then equivalence of all factor loadings




Figure 1. Path diagram with the structured means in group 1.

is imposed across groups (*weak factorial invariance*). The model fit of the weak invariance model is then compared to that of the configural invariance model. If the added equality constraints do not substantially decrease the model fit, then equivalence of covariance and/or mean structures can be imposed across groups. Otherwise, constraints on factor loadings that caused the lack of fit should be removed, not simultaneously but one at a time, until partial invariance of factor loadings is established. Once the (partial) weak invariance model is established, researchers can further examine the equivalence of means, covariances, or both, depending on their research questions. Equivalence of mean structures can be explored by invariance tests on the intercepts for observed measures and factors. Invariance of covariance structures involves two levels which can be established with or without the mean structures. First, one can test the invariance of the factor covariance matrix (*strong factorial invariance*). Then, if strong factorial invariance is satisfied, one can proceed to the test of the invariance of covariance structures of measurement errors (*strict factorial invariance*).

In the following, an empirical example is used to illustrate the measurement invariance testing procedure. This example focuses on equivalence of factor loadings and mean structures for observed measures and factors between two groups, which

Table 1. Creating two separate data sets in the type of covariance matrix in the SAS system

```

data group1 (type=cov);
infile datalines missover;
input _NAME_ $ _TYPE_ $ V1-V6;
datalines;
. MEAN 49.14 82.60 104.95 78.58 54.95 119.91
V1 COV 154.54
V2 COV 44.75 90.23
V3 COV 40.98 22.77 78.76
V4 COV 41.35 2.83 7.92 220.12
V5 COV 23.28 9.12 0.75 61.46 159.88
V6 COV 47.08 22.88 16.05 125.08 84.31 332.26
;
data group2 (type=cov);
infile datalines missover;
input _NAME_ $ _TYPE_ $ V1-V6;
datalines;
. MEAN 55.01 81.22 97.01 72.67 46.57 124.28
V1 COV 124.93
V2 COV 52.19 80.67
V3 COV 64.45 42.85 83.56
V4 COV 59.95 33.10 38.34 290.65
V5 COV 32.32 16.09 18.29 124.71 169.17
V6 COV 87.15 39.70 51.82 174.20 108.39 355.22
;
run;

```

correspond to the tenth model from the taxonomy of 13 partially nested models operationalized by Marsh *et al.* (2009, Table 1, p. 443).

3. Example data

This example data set discussed here is taken from Thompson and Green (2006, p. 139, Table 5.2, Dataset 2), which contains six measured variables aiming to assess preschool children's academic (V1-V3) and social school readiness (V4-V6). Preschool children were divided into two groups: day-care (group 1) and home-care (group 2). To conduct tests of measurement invariance comparing the two groups, we need the means and covariance matrix of the study variables (V1-V6) for each group. The sample sizes are 250 and 150 for group 1 and group 2, respectively. The model for each group is a confirmatory factor model with two correlated factors (F1 and F2). F1 has the first three indicators (V1-V3) and F2 has the last three (V4-V6). Figure 1 shows the path diagram with the mean structures in group 1 (see also Thompson & Green, 2006, Figure 5.2, p. 132). Note that there is variation among the path diagrams for the same model due to the different drawing conventions. McDonald and Ho (2002) discussed three types of conventions, and asserted that there is no 'right' way to draw a path diagram. One drawing convention might seem logical to some users, but confusing to others. In this example, we choose to use Convention 1 described in McDonald and Ho (2002) because the translation from the path diagram to the programming of the model specification is most straightforward in PROC TCALIS. This point will be further discussed in the next section.

To read in the means and covariance matrices in SAS, users need to explicitly specify the data type as covariance matrix by adding the data set option in parentheses,

type = cov, in the DATA step (Table 1). Though only two groups are illustrated, it is very easy to generalize the procedures to cases with more than two groups.

3.1. SAS modelling languages

In the TCALIS procedure, there is a collection of modelling languages available to specify models (e.g., LINEQS, LISMOD, and PATH), each reflecting different modelling terminology and philosophies. Traditionally, most users chose the LINEQS language in the old CALIS procedure (e.g., Brown, 2006), and this is still available in PROC TCALIS. Also available is the LISMOD language, developed for users familiar with LISREL models (Jöreskog & Sörbom, 1996). The PATH language seems to be the most flexible, and it was used for this example. Other modelling languages are also provided (e.g., FACTOR and MSTRUCT), but they are mostly designed for specific purposes, thus omitted in this illustration. Essentially, different modelling languages are different ways of doing the same analyses.

For simplicity, a model specified using the PATH language is called a PATH model. A complete model specification usually needs three major statements (i.e., PATH, PVAR, and PCOV) plus a MEAN statement if mean structures are involved.

By using Convention 1 (McDonald & Ho, 2002) to draw the path diagram, a direct translation from the figure to the PATH model is very conveniently achieved by following the rules below:

- Each single-headed arrow in the path diagram is specified in the PATH statement.
- Each double-headed arrow pointing to a single variable is specified in the PVAR statement.
- Each double-headed arrow pointing to two different variables is specified in the PCOV statement.

3.2. General description of the stepwise procedure

The stepwise procedure suggested by Thompson and Green (2006, Table 5.4, pp. 144–147) is used to examine the between-group difference in factor means under partial invariance so that the selected model fit indices – chi-square (χ^2), standardized root mean square residual (SRMR), and root mean square error of approximation (RMSEA) – reported by PROC TCALIS can be validated. The metric in this example is defined by fixing the variance of factors to 1 in group 1 and imposing between-group equality constraints of factor loadings (except in step 1); by doing so, we are able to evaluate all between-group constraints on factor loadings. Decision rules between steps are such that, except for step 1, the χ^2 difference test, together with SRMR and RMSEA, is used to assess differential fit of nested models; and the decision to remove individual between-group constraints is based on the modification indices (MI), also known as Lagrange multiplier (LM) tests, in order to improve the model fit (Chou & Bentler, 1990).

The χ^2 difference test is not considered best practice by some researchers when mean structures get involved because it is strongly influenced by sample size, and alternative criteria have been proposed (Chen, 2007; Cheung & Rensvold, 2002; Fan & Sivo, 2009). Consequently, different criteria can lead to different decisions. In this tutorial, we follow the steps presented by Thompson and Green (2006, Table 5.4, pp. 144–147). Details of alternative tests for nested model evaluation are not discussed here, but readers can find related information from the references cited above or in the wider literature.

Table 2. Step 1: Evaluating the same factor model in each of the groups and in the groups combined

```

    *** Group 1 ***;
1  proc tcalis data=group1 cov method=ml nobs=250;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  path
4      F1 -> V1 b1,
5      F1 -> V2 b2,
6      F1 -> V3 b3,
7      F2 -> V4 b4,
8      F2 -> V5 b5,
9      F2 -> V6 b6;
10 pvar
11     F1 = 1.0,
12     F2 = 1.0,
13     V1-V6 = e1-e6;
14 pcov
15     F1 F2 = d12;
16 run;

    *** Group 2 ***;
17 proc tcalis data=group2 cov method=ml nobs=150;
18 fitindex NOINDEXTYPE on(only) = [CHISQ DF PROBCHI SRMSR RMSEA];
19 path
20     F1 -> V1 b1,
21     F1 -> V2 b2,
22     F1 -> V3 b3,
23     F2 -> V4 b4,
24     F2 -> V5 b5,
25     F2 -> V6 b6;
26 pvar
27     F1 = 1.0,
28     F2 = 1.0,
29     V1-V6 = e1-e6;
30 pcov
31     F1 F2 = d12;
32 run;

    *** unconstrained multigroup ***;
33 proc tcalis method =ml;
34 fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
35 group 1 / data=group1 nobs=250;
36 group 2 / data=group2 nobs=150;
37 model 1 / group=1;
38 path
39     F1 -> V1 b1,
40     F1 -> V2 b2,
41     F1 -> V3 b3,
42     F2 -> V4 b4,
43     F2 -> V5 b5,
44     F2 -> V6 b6;

```

Continued

Table 2. (Continued)

```

45     pvar
46     F1 = 1.0,
47     F2 = 1.0,
48     V1-V6 = e1-e6;
49     pcov
50     F1 F2 = d12;
51     model 2 / group=2;
52     refmodel 1 / AllNewParms;
53     run;

```

3.3. Step 1: Evaluating single-group and unconstrained multigroup factor models for configural invariance

In step 1, the same factor model (Figure 1) is evaluated in each group and in the groups combined for configural invariance. The TCALIS procedure is invoked by the PROC TCALIS statement, and data set information and estimation method are specified by using options in this statement. The DATA = option tells SAS which data set is used. For example, in Table 2, line 1, the single-group model for group 1 uses the SAS data set 'group1', and because the type of this data set is covariance, the COV and NOBS = options are needed, that is, PROC TCALIS will analyse a covariance matrix instead of a raw data set, and it is based on 250 observations. The METHOD = option specifies the estimation method. The normal-theory maximum-likelihood method is used throughout this example.

In the PATH statement, all single-headed arrows are translated into the corresponding path entries. An arrow symbol, \rightarrow (or \leftarrow), is used between two variables. If A is the antecedent variable to B, then the order of A and B in the path entry is interchangeable as long as the head of the arrow points to the outcome variable, B. In other words, $A \rightarrow B$ is equivalent to $B \leftarrow A$. A fixed value or a unique parameter name is also specified for each path entry. Fixed values represent fixed model parameters, whereas a parameter name tells SAS that this particular parameter will be freely estimated in the model. Path entries are separated by a comma. For example, the factor loadings from F1 to V1, V2, and V3 are specified as b1, b2, and b3, respectively (see Table 2, lines 4-6); and those from F2 to V4, V5, and V6 are b4, b5, and b6, respectively (see lines 7-9), in the single-group model for group 1. The multigroup model in lines 39-44 specifies the same path entries in group 1 as those in the single-group models. The path specification for group 2 in the multigroup model will be discussed later.

In the PVAR statement, all double-headed arrows pointing to a single variable, representing variance parameters for factor disturbances and measurement errors, are translated to corresponding variance entries. One or multiple variables are listed on the left-hand side of an equality sign, and one or multiple fixed values and/or unique parameter names are listed on the right-hand side. Similarly, a fixed value represents a fixed parameter, whereas a parameter name represents a freely estimated parameter. Variance entries are separated by a comma. For example, in Table 2, lines 11-13, three variance entries are specified in single-group models for group 1, that is, variances of F1 and F2 are set to 1 in order to define the metric, and six error variances are defined as freely estimated parameters, labelled as e1 to e6 associated with measured variables V1 to V6, separately. The syntax V1-V6 is a handy shortcut to represent the list from V1 to V6, as is the syntax e1-e6. Users can always type all six entries

($V_1=e_1$, $V_2 = e_2$, . . . , $V_6=e_6$) to achieve the same thing. The multigroup model in lines 46–48 specifies the same variance entries in group 1 as those in single-group models. The variance specification for group 2 in multigroup model is also going to be discussed later.

In the PCOV statement, all double-headed arrows pointing to two different variables, representing covariance parameters, are translated into covariance entries. A pair of variables separated by a space is listed on the left-hand side of an equality sign, and a fixed value or a unique parameter name is listed on the right-hand side. The usage of fixed values and parameter names is the same as in the PATH and PVAR statement. Different covariance entries are separated by a comma. For example, there is only one pair of correlated disturbances in each model, and thus a single covariance entry is specified in single-group model for group 1 (see Table 2, line 15). The covariance parameter is labelled as d12. The multigroup model in Table 2 specifies the same covariance entry for group 1 as in the single-group model.

We now explain the model specification of group 2 in the multigroup model in Table 2. First, there are two GROUP statements, each designating a group number. Here, groups 1 and 2 are defined (see line 35 and 36). After the slash, the DATA = and NOBS = options in the GROUP statement indicate the data and the number of observations for each group.

Second, there are two MODEL statements (see lines 37 and 51), each signifying the beginning of a model specification block and designating a number for the model. Until another MODEL statement is encountered, the PATH, PVAR, and PCOV statements in between belong to the precedent model specification block. For example, lines 38–50 constitute a complete specification block, and this PATH model is designated as model specification 1. The GROUP = option can list one or multiple integers which represent the groups to be fitted by the model. For example, the statement (model 5/group = 6 7;) means that model specification 5 fits groups 6 and 7 (if defined in previous GROUP statements).

Third, the REFMODEL statement (see line 52), as a referencing tool, is used to build the model specification within the scope of the MODEL statement in line 51. It tells SAS that the pattern of fixed and free parameters in model specification 2 is equivalent to model specification 1. The AllNewParms option, standing for ‘all new parameters’, means that the parameter names are different from those in model specification 1. By default, SAS adds a suffix, `_md12`, to the corresponding parameter names in model specification 1 so as to create unique names in model specification 2. With unique names in the combined groups, there is no between-group constraint.

From the output, the unconstrained configural model fits the data adequately ($\chi^2(16) = 17.79$, $p = .336$, SRMR = .037, RMSEA = .024). We now proceed to the next step to evaluate the weak factorial invariance.

3.4. Step 2a: Evaluating between-group equivalence of factor loadings for weak factorial invariance

In Table 3, constraints on all factor loadings are placed between groups, and other parameters are freely estimated in each group. This new model in group 2 is specified by integrating some entry replacements in lines 21–26 into the referencing model referred to by the REFMODEL statement in line 20. In other words, the new entries in the PVAR and PCOV statements, nested within the scope of the REFMODEL statement, designate new parameter names in group 2 by replacing those in previous PVAR and PCOV statements in group 1 (lines 13–18); and the absence of new entries for the rest of the parameters implies between-group constraints. In lines 22–24 and 26, new parameter names are

Table 3. Step 2a: Evaluating between-group equivalence of factor loadings

```

1  proc tcalis method=ml modification;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;

6  path
7    F1 -> V1 b1,
8    F1 -> V2 b2,
9    F1 -> V3 b3,
10   F2 -> V4 b4,
11   F2 -> V5 b5,
12   F2 -> V6 b6;

13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;

17  pcov
18   F1 F2 = d12;

19  model 2 / group=2;
20   refmodel 1;

21  pvar
22   F1 = g2_d1,
23   F2 = g2_d2,
24   V1-V6 = g2_e1-g2_e6;

25  pcov
26   F1 F2 = g2_d12;

27  run;

```

assigned by adding a prefix, *g2_*, to the old parameter names, that is, the variances of F1, F2, V1, V2, . . . , and V6 are *g2_d1*, *g2_d2*, *g2_e1*, *g2_e2*, . . . , and *g2_e6*, respectively, and the covariance of F1 and F2 is *g2_d12*. The rest of the parameters in group 2 share the same names with those in group 1, and thus, between-group constraints are imposed. The MODIFICATION option (see line 1) was included to request MI/LM tests on the constraints imposed on parameters in the two groups.

Following Thompson and Green (2006), multiple nested models are evaluated by the χ^2 difference test in combination with the changes in SRMR and RMSEA. Fit indices ($\chi^2(20) = 26.07$, $p = .164$, SRMR = .048, RMSEA = .039) are given in the SAS output, and thus the associated difference tests between the configural invariance (step 1) and the weak invariance (step 2a) models can be calculated ($\Delta\chi^2(4) = 8.28$, $p = .082$, Δ SRMR = .011, Δ RMSEA = .015). As noted by Thompson and Green (2006), there is room for mixed opinions on model fit, so that arbitrary decisions might be made given these fit indices. As noted before, we follow the decisions made by Thompson and Green (2006) throughout our example so that the models are identical to theirs, and therefore the comparison of the fit indices is meaningful. Here, we follow the conclusion that the change in fit is non-minimal, and the constraints on loadings that produced the lack of fit need to be removed.

Now, careful readers may find discrepancies in RMSEA between SAS and those from EQS reported by Thompson and Green. Similarly, discrepancies in SRMR will also be

found in subsequent steps. Reasons of such inconsistencies between SAS and EQS will be discussed in the Section 4.

3.5. Step 2b: Evaluating between-group equivalence of individual loadings for partial weak factorial invariance

In this step, constraints on loadings that produced the lack of fit are to be identified and removed. Following Thompson and Green (2006), we used LM/MI tests to identify implausible equality constraints and adopted their decisions (see also Saris, Satorra, & Van der Veld, 2009, for an in-depth discussion of LM/MI tests). In practice, one should decide on theoretical grounds which constraints should be removed. The constraint placed on loadings for V3 in step 2a gives the largest value on the LM/MI test, so this constraint was removed first. To remove a certain loading constraint, a new path entry is needed to replace the old path entry. In Table 4 (step 2b, substep 1), the PATH statement in lines 21 and 22, within the scope of the REFMODEL statement in line 20, specifies a new path entry, which tells SAS the loading from F1 to V3 is g2_b3 in group 2. By designating a new name to the factor loading in group 2, the constraint on loadings for V3 is removed. The use of the REFMODEL statement is not mandatory, if the user chooses to write the complete model specification in group 2. That is, users can retype all entries in group 1 assigning identical or different parameter names in group 2, depending on whether the between-group constraints are imposed or not. Usually, using the REFMODEL statement can save the user some programming effort when model specification is very complicated.

Fit indices ($\chi^2(19) = 22.04$, $p = .282$, SRMR = .042, and RMSEA = .028) from the output reveal that the improvement in fit is marginal by comparing the models from step 2a and step 2b, substep 1 ($\Delta\chi^2(1) = 4.03$, $p = .045$, Δ SRMR = .006, Δ RMSEA = .011). Following Thompson and Green (2006), in order to avoid misspecification, the loadings for V3 will be allowed to differ in all future analyses. Meanwhile, the loading constraint for V6 gives the largest value in the LM/MI tests, and this constraint is removed.

In step 2b, substep 2, two path entries are specified in model specification 2 to remove constraints on loadings for V3 and V6 in lines 50–52. From the SAS output, the fit indices for this model ($\chi^2(18) = 19.16$, $p = .382$, SRMR = .040, and RMSEA = .018) indicate that the fit is improved, but to a lesser degree, compared to the model in step 2b, substep 1 ($\Delta\chi^2(1) = 2.88$, $p = .090$, Δ SRMR = .002, Δ RMSEA = .010). Going with the decision of Thompson and Green, the loading for V6 will be allowed to differ in all future analyses to avoid misspecification. Meanwhile, the loading constraint for V1 gives the largest value in the LM/MI tests, and this constraint is removed next.

In step 2b, substep 3, the constraint on the loadings for V1 is further removed in line 81. This model fit looks good ($\chi^2(17) = 17.86$, $p = .398$, SRMR = .037, RMSEA = .016), but the improvement in fit is not substantial ($\Delta\chi^2(1) = 1.30$, $p = .254$, Δ SRMR = .003, Δ RMSEA = .002), compared to the model in step 2b, substep 2. We will follow the arbitrary choice in Thompson and Green (2006). That is, because the fit is improved only minimally, the constraint on the loadings for V1 is not removed. So far, the constraints on loadings for V3 and V6 have been removed, and partial invariance for factor loadings has been established.

3.6. Step 3a: Evaluating between-group equivalence of intercepts for measured variables

When mean structures are introduced, the multigroup analyses become more complicated. In the MEAN statement, entries of intercepts for measures and factors are

Table 4. Step 2b: Evaluating between-group equivalence of individual loadings

```

*** Step 2b, Substep 1: remove the constraint on loadings for V3 ***;
1  proc tcalis method=ml modification;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;
6  path
7    F1 -> V1  b1,
8    F1 -> V2  b2,
9    F1 -> V3  b3,
10   F2 -> V4  b4,
11   F2 -> V5  b5,
12   F2 -> V6  b6;
13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;
17  pcov
18   F1 F2 = d12;
19  model 2 / group=2;
20   refmodel 1;
21  path
22   F1 -> V3  g2_b3;
23  pvar
24   F1 = g2_d1,
25   F2 = g2_d2,
26   V1-V6 = g2_e1-g2_e6;
27  pcov
28   F1 F2 = g2_d12;
29  run;

*** Step 2b, Substep 2: remove the constraint on loadings for V6 ***;
30  proc tcalis method=ml modification;
31  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
32  group 1 / data=group1 nobs=250;
33  group 2 / data=group2 nobs=150;
34  model 1 / group=1;
35  path
36   F1 -> V1  b1,
37   F1 -> V2  b2,
38   F1 -> V3  b3,
39   F2 -> V4  b4,
40   F2 -> V5  b5,
41   F2 -> V6  b6;
42  pvar
43   F1 = 1.0,
44   F2 = 1.0,
45   V1-V6 = e1-e6;
46  pcov
47   F1 F2 = d12;

```

Continued

Table 4. (Continued)

```

48 model 2 / group = 2;
49   refmodel 1;
50 path
51   F1 -> V3 g2_b3,
52   F2 -> V6 g2_b6;
53 pvar
54   F1 = g2_d1,
55   F2 = g2_d2,
56   V1-V6 = g2_e1-g2_e6;
57 pcov
58   F1 F2 = g2_d12;
59 run;

   *** Step 2b, Substep 3: remove the constraint on loadings for V1 ***;
60 proc tcalis method=ml modification;
61 fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
62 group 1 / data=group1 nobs=250;
63 group 2 / data=group2 nobs=150;
64 model 1 / group=1;
65 path
66   F1 -> V1 b1,
67   F1 -> V2 b2,
68   F1 -> V3 b3,
69   F2 -> V4 b4,
70   F2 -> V5 b5,
71   F2 -> V6 b6;
72 pvar
73   F1 = 1.0,
74   F2 = 1.0,
75   V1-V6 = e1-e6;
76 pcov
77   F1 F2 = d12;
78 model 2 / group = 2;
79   refmodel 1;
80 path
81   F1 -> V1 g2_b1,
82   F1 -> V3 g2_b3,
83   F2 -> V6 g2_b6;
84 pvar
85   F1 = g2_d1,
86   F2 = g2_d2,
87   V1-V6 = g2_e1-g2_e6;
88 pcov
89   F1 F2 = g2_d12;
90 run;

```

specified. One or more variables can be listed on the left-hand side of an equality sign, and correspondingly, one or more fixed values and/or unique parameter names need to be listed on the right-hand side. Entries are separated by a comma. For example, in Table 5, intercepts for factors are specified in group 1 and group 2, separately. That is,

Table 5. Step 3a: Evaluating between-group equivalence of intercepts for measures

```

1  proc tcalis method=ml modification maxiter=300 maxfunc=300;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;
6  path
7    F1 -> V1  b1,
8    F1 -> V2  b2,
9    F1 -> V3  b3,
10   F2 -> V4  b4,
11   F2 -> V5  b5,
12   F2 -> V6  b6;
13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;
17  pcov
18   F1 F2 = d12;
19  mean
20   F1 F2 = 0 0,
21   V1-V6 = aV1-aV6;
22  model 2 / group = 2;
23   refmodel 1;
24  path
25   F1 -> V3  g2_b3,
26   F2 -> V6  g2_b6;
27  pvar
28   F1 = g2_d1,
29   F2 = g2_d2,
30   V1-V6 = g2_e1-g2_e6;
31  pcov
32   F1 F2 = g2_d12;
33  mean
34   F1 F2 = g2_aF1  g2_aF2,
35   V3 V6 = g2_aV3  g2_aV6;
36  run;

```

in group 1, intercepts for F1 and F2 are set to 0 in line 20, and intercepts for V1, V2, V4, and V5 are named as aV1, aV2, aV4, and aV5 in line 21; whereas in group 2, intercepts for F1, F2, V1, V2, V4, and V5 are freely estimated by assigning different parameter names in lines 34–35: g2_aF1, g2_aF2, g2_aV1, g2_aV2, g2_aV4, and g2_aV5.

The SAS output reports that $\chi^2(20) = 54.44$, $p < .001$, SRMR = .060, and RMSEA = .093, and the difference in fit is significant ($\Delta\chi^2(2) = 35.28$, $p < .001$, $\Delta\text{SRMR} = .020$, $\Delta\text{RMSEA} = .075$) compared to the model in step 2b, substep 2. We conclude that model fit differs significantly. By examining the LM/MI tests, we are again going with the arbitrary decision that the constraint on the intercepts for V1 will be removed.

3.7. Step 3b: Evaluating between-group equivalence of intercepts for individual measured variables

In step 3b, substep 1 (Table 6), the same logic and decision rules are applied to evaluate the constraint on intercepts for V1. The improvement in fit is considered significant, and the intercept for V1 is allowed to differ in all future analyses. By examining the LM/MI tests, we will remove the constraint on the intercepts for V4.

In step 3b, substep 2 (Table 6), the same logic and decision rules are applied to evaluate the constraint on intercepts for V4. The improvement in fit is considered significant, and the intercept for V4 is allowed to differ in all future analyses. So far, the constraints on intercepts for V1 and V4 have been removed, and we assume the remaining constraints are appropriate.

3.8. Step 4a: Evaluating between-group equivalence of intercepts for factors

In step 4a (Table 7), no entry is specified for intercepts for both factors (F1 and F2) in the specification block in lines 24–34, meaning that each of the intercepts is equated between groups. Comparing the models from step 3b, substep 2, to step 4a shows that model fit differs significantly, and therefore one of the constraints will be removed.

3.9. Step 4b: Evaluating between-group equivalence of intercepts for individual factors

In step 4b, substep 1 (Table 8), we test the equivalence of intercepts for F1 by removing the constraint on the intercepts for F2. In line 34, a new parameter name, g2_aF2, is assigned to the intercept for F2 in group 2. By comparing the fit of this model to that of step 3b, substep 2, we conclude that the improvement is not considered significant.

In step 4b, substep 2 (Table 8), we test the equivalence of intercepts for F2 by removing the constraint on the intercepts for F1. In line 70, a new parameter name, g2_aF1, is assigned to the intercept for F1 in group 2. By comparing the fit of this model to that of step 3b, substep 2, we conclude that the improvement is significant, and that the means on the second factor differ substantially in these two groups.

As noted before, when mean structures are involved, multigroup SEM grows more complicated. As a result, non-convergence may occur in the estimation process. One possible solution is to increase both the maximum number of iterations and function calls. The MAXITER = and MAXFUNC = options in the PROC TCALIS statement allow the user to set different numbers of iterations and function calls (e.g., in line 1 of Table 5–Table 8). An alternative is to provide starting values. In a PATH model, adding a value in paired parentheses after the parameter name tells SAS to use this value as the starting value. For example, the PATH entry

```
F1- > V1  b1  (0.5)
```

means that PROC TCALIS will use 0.5 as the starting value to estimate the parameter b1.

The use of the FITINDEX statement (e.g., in line 2 of Table 2–Table 8) is optional. In this example, the NOINDEXTYPE option disables the display of index types in the fit summary table, and the ON(ONLY) = option turns on the printing of χ^2 , degrees of freedom, *p*-value, SRMR, and RMSEA. If the FITINDEX statement is omitted, a complete fit summary table is displayed.

Table 6. Step 3b: Evaluating between-group equivalence of individual intercepts for measures

```

*** Step 3b, Substep 1: remove the constraint on intercepts for V1 ***;
1  proc tcalis method=ml modification maxiter=300 maxfunc=300;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;
6  path
7    F1 -> V1 b1,
8    F1 -> V2 b2,
9    F1 -> V3 b3,
10   F2 -> V4 b4,
11   F2 -> V5 b5,
12   F2 -> V6 b6;
13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;
17  pcov
18   F1 F2 = d12;
19  mean
20   F1 F2 = 0 0,
21   V1-V6 = aV1-aV6;
22  model 2 / group = 2;
23   refmodel 1;
24  path
25   F1 -> V3 g2_b3,
26   F2 -> V6 g2_b6;
27  pvar
28   F1 = g2_d1,
29   F2 = g2_d2,
30   V1-V6 = g2_e1-g2_e6;
31  pcov
32   F1 F2 = g2_d12;
33  mean
34   F1 F2 = g2_aF1 g2_aF2,
35   V1 V3 V6 = g2_aV1 g2_aV3 g2_aV6;
36  run;

*** Step 3b, Substep 2: remove the constraint on intercepts for V4 ***;
37  proc tcalis method=ml modification maxiter=300 maxfunc=300;
38  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
39  group 1 / data=group1 nobs=250;
40  group 2 / data=group2 nobs=150;
41  model 1 / group=1;
42  path
43   F1 -> V1 b1,
44   F1 -> V2 b2,
45   F1 -> V3 b3,
46   F2 -> V4 b4,
47   F2 -> V5 b5,
48   F2 -> V6 b6;

```

Continued

Table 6. (Continued)

```

49  pvar
50    F1 = 1.0,
51    F2 = 1.0,
52    V1-V6 = e1-e6;
53  pcov
54    F1 F2 = d12;
55  mean
56    F1 F2 = 0 0,
57    V1-V6 = aV1-aV6;
58  model 2 / group = 2;
59  refmodel 1;
60  path
61    F1 -> V3  g2_b3,
62    F2 -> V6  g2_b6;
63  pvar
64    F1 = g2_d1,
65    F2 = g2_d2,
66    V1-V6 = g2_e1-g2_e6;
67  pcov
68    F1 F2 = g2_d12;
69  mean
70    F1 F2 = g2_aF1 g2_aF2,
71    V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
72  run;

```

4. Results

Model fit indices from all steps are gathered together in Table 9, and difference tests of fit indices from SAS comparing nested models are summarized in Table 10. As noted by Thompson and Green (2006, p. 156), fit indices produced by EQS were reported in the chapter. For single-group models, the values of SRMR and RMSEA in EQS and SAS 9.2 are identical. However, for multigroup models, different values of SRMR are reported by SAS and EQS. Some of the differences are not trivial (steps 3a, 4a, and 4b, substep 2). In PROC TCALIS, the SRMR for each group is calculated first, and then the group weight,

$$a_r = \frac{N_r - 1}{N - k},$$

is used to calculate the overall SRMR, that is,

$$\text{overall SRMR} = \sqrt{\sum_{r=1}^k a_r \text{SRMR}_r^2},$$

where SRMR_r is the SRMR for group r , N_r is the sample size of group r , N is the total sample size, and k is the number of groups. If the FITINDEX statement is omitted, the user can find the SRMR for each group in the complete fit summary table. In contrast, there are two reasons why EQS gives different SRMRs. First, when there is mean structure, EQS bases the calculation on covariances only, and ignores mean residuals. Second, EQS

Table 7. Step 4a: Evaluating between-group equivalence of intercepts for factors

```

1  proc tcalis method=ml modification maxiter=300 maxfunc=300;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;
6  path
7    F1 -> V1  b1,
8    F1 -> V2  b2,
9    F1 -> V3  b3,
10   F2 -> V4  b4,
11   F2 -> V5  b5,
12   F2 -> V6  b6;
13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;
17  pcov
18   F1 F2 = d12;
19  mean
20   F1 F2 = 0 0,
21   V1-V6 = aV1-aV6;
22  model 2 / group = 2;
23   refmodel 1;
24  path
25   F1 -> V3  g2_b3,
26   F2 -> V6  g2_b6;
27  pvar
28   F1 = g2_d1,
29   F2 = g2_d2,
30   V1-V6 = g2_e1-g2_e6;
31  pcov
32   F1 F2 = g2_d12;
33  mean
34   V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
35  run;

```

calculates the overall SRMR in a direct way, that is, the squared residuals in each group are added up and the sum is divided by $kp(p + 1)$, where k is the number of groups, and p is the number of observed variables in a group (David Sookne, personal communication, August 2010). The discrepancies in the values of RMSEA result from the adjustment for multigroup models proposed by Steiger (1998), that is, an averaged degrees of freedom should be employed in the calculation of the RMSEA. As Thompson and Green (2006, p. 158) noted, EQS had not yet incorporated Steiger's correction at the time of their writing, but EQS 6.1 did so. In this example, after multiplying the RMSEA values reported by EQS by $\sqrt{2}$, they are the same as those reported by SAS. As a cross-validation, we also fit all the example models in *Mplus* 5.21, and the values of SRMR and RMSEA agree perfectly with SAS 9.2. Many other fit indices are available in PROC TCALIS, such as the

Table 8. Step 4b: Evaluating between-group equivalence of individual intercepts for factors

```

    *** Step 4b, Substep 1: test of intercepts between groups for F1 ***;
1  proc tcalis method=ml modification maxiter=300 maxfunc=300;
2  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
3  group 1 / data=group1 nobs=250;
4  group 2 / data=group2 nobs=150;
5  model 1 / group=1;
6  path
7    F1 -> V1 b1,
8    F1 -> V2 b2,
9    F1 -> V3 b3,
10   F2 -> V4 b4,
11   F2 -> V5 b5,
12   F2 -> V6 b6;
13  pvar
14   F1 = 1.0,
15   F2 = 1.0,
16   V1-V6 = e1-e6;
17  pcov
18   F1 F2 = d12;
19  mean
20   F1 F2 = 0 0,
21   V1-V6 = aV1-aV6;
22  model 2 / group = 2;
23   refmodel 1;
24  path
25   F1 -> V3 g2_b3,
26   F2 -> V6 g2_b6;
27  pvar
28   F1 = g2_d1,
29   F2 = g2_d2,
30   V1-V6 = g2_e1-g2_e6;
31  pcov
32   F1 F2 = g2_d12;
33  mean
34   F2 = g2_aF2,
35   V1 V3 V4 V6 = g2_aV1 g2_aV3 g2_V4 g2_aV6;
36  run;

    *** Step 4b, Substep 2: test of intercepts between groups for F2 ***;
37  proc tcalis method=ml modification maxiter=300 maxfunc=300;
38  fitindex NOINDEXTYPE on(only)=[CHISQ DF PROBCHI SRMSR RMSEA];
39  group 1 / data=group1 nobs=250;
40  group 2 / data=group2 nobs=150;
41  model 1 / group=1;
42  path
43   F1 -> V1 b1,
44   F1 -> V2 b2,
45   F1 -> V3 b3,
46   F2 -> V4 b4,
47   F2 -> V5 b5,
48   F2 -> V6 b6;

```

Continued

Table 8. (Continued)

```

49  pvar
50    F1 = 1.0,
51    F2 = 1.0,
52    V1-V6 = e1-e6;
53  pcov
54    F1 F2 = d12;
55  mean
56    F1 F2 = 0 0,
57    V1-V6 = aV1-aV6;
58  model 2 / group = 2;
59    refmodel 1;
60  path
61    F1 -> V3  g2_b3,
62    F2 -> V6  g2_b6;
63  pvar
64    F1 = g2_d1,
65    F2 = g2_d2,
66    V1-V6 = g2_e1-g2_e6;
67  pcov
68    F1 F2 = g2_d12;
69  mean
70    F1 = g2_aF1,
71    V1 V3 V4 V6 = g2_aV1  g2_aV3  g2_V4  g2_aV6;
72  run;

```

comparative fit index and the non-normal fit index (sometimes known as the Tucker-Lewis index). Technical details of these fit indices in PROC TCALIS can be found in the SAS/STAT 9.2 User's Guide (SAS Institute, 2008, pp. 6892-6905).

Parameter estimates reported by PROC TCALIS are identical to those obtained by EQS (Thompson & Green, 2006, Table 5.5, p. 150) and are omitted here.

5. Comments and conclusions

In this tutorial, we have illustrated the steps involved in testing the measurement invariance on factor means using SAS PROC TCALIS. As described above, invariance can also be tested at more stringent levels, such as factor variances and covariances, and possibly error variances. Although space has not permitted complete illustrations of all of them, any invariance test can be easily carried out as long as one follows the general principle such that, for the parameters in one group, identical names in other groups imply constraints while different names designate freely estimated parameters across groups.

We have confined ourselves to illustrating the PATH language, given its flexibility. Other modelling languages can also be used to specify the models in this illustration. Users with knowledge of EQS or LISREL will find it easy to specify models by using the LINEQS or LISMOD language. In fact, the output templates associated with different modelling languages (PATH, LINEQS, and LISMOD) are similar to those generated by specialized SEM software packages (*Mplus*, EQS, and LISREL).

In addition to the normal-theory maximum likelihood method, other estimation methods are also available. For example, METHOD = WLS requests the weighted least squares method. The user needs to provide a non-singular weight matrix in the

Table 9. Selected model fit index results from SAS 9.2 and Thompson and Green (2006)

	Chi-square		SRMR		RMSEA		
	Degrees of freedom	Thompson and Green	SAS 9.2	Thompson and Green	SAS 9.2	Thompson and Green	SAS 9.2
Step 1: group 1	8	8.02	8.02	.031	.031	.003	.003
Step 1: group 2	8	9.77	9.77	.045	.045	.039	.039
Step 1: groups 1 and 2 combined	16	17.79	17.79	.039	.037	.017	.024
Step 2a: place constraints on all loadings	20	26.07	26.07	.048	.048	.028	.039
Step 2b, substep 1: remove constraint on loadings for V3	19	22.04	22.04	.043	.042	.020	.028
Step 2b, substep 2: remove constraint on loadings for V6	18	19.16	19.16	.041	.040	.013	.018
Step 2b, substep 3: remove constraint on loadings for V1	17	17.86	17.86	.039	.037	.011	.016
Step 3a: place constraints on all intercepts except for V3 and V6	20	54.44	54.44	.053	.060	.066	.093
Step 3b, substep 1: remove constraint on intercepts for V1	19	29.43	29.43	.047	.044	.037	.053
Step 3b, substep 2: remove constraint on intercepts for V4	18	19.16	19.16	.041	.035	.013	.018
Step 4a: place constraints on intercepts for both factors	20	56.39	56.39	.061	.092	.068	.096
Step 4b, substep 1: test the constraint of intercepts for F1	19	21.28	21.28	.041	.039	.017	.025
Step 4b, substep 2: test the constraint of intercepts for F2	19	56.17	56.17	.060	.090	.070	.099

Note. SRMR = standardized root mean square residual. RMSEA = root mean square error of approximation. V standards for variable and F standards for factor.

INWGT = option in PROC TCALIS. If no weight matrix is provided, the WLS method uses the inverse matrix of estimated asymptotic covariances of the sample covariance or correlation matrix as the weight matrix. In this case, the WLS method is equivalent to Browne's asymptotically distribution-free estimation (Browne, 1982, 1984).

Though the TCALIS procedure is a great improvement over the old CALIS procedure, there are still some noticeable limitations. For example, PROC TCALIS is not capable

Table 10. Difference tests of fit indices summarized from SAS 9.2 results

Steps to be compared			$\Delta\chi^2$	Δdf	p -value	$\Delta SRMR$	$\Delta RMSEA$
Step 1	VS	Step 2a	8.28	4	.082	.011	.015
Step 2a	VS	Step 2b, substep 1	4.03	1	.045	.006	.011
Step 2b, substep 1	VS	Step 2b, substep 2	2.88	1	.090	.002	.010
Step 2b, substep 2	VS	Step 2b, substep 3	1.30	1	.254	.003	.002
Step 2b, substep 2	VS	Step 3a	35.28	2	<.001	.020	.075
Step 3a	VS	Step 3b, substep 1	25.01	1	<.001	.016	.041
Step 3b, substep 1	VS	Step 3b, substep 2	10.27	1	.001	.009	.035
Step 3b, substep 2	VS	Step 4a	37.23	2	<.001	.057	.078
Step 3b, substep 2	VS	Step 4b, substep 1	2.12	1	.145	.004	.007
Step 3b, substep 2	VS	Step 4b, substep 2	37.02	1	<.001	.055	.081

Note. Δ indicates difference in an index between two models.

of analysing ordered-categorical data or data with nested structures (multilevel SEM). Currently, to the knowledge of authors, *Mplus* appears to be a better choice for SEM with ordered-categorical data (Lubke & Muthén, 2004; Millsap & Yun-Tein, 2004).

In summary, this tutorial shows that the TCALIS procedure extends the scope of SEM models that can be handled by SAS, that is, multiple-group analysis. The fit indices (e.g., χ^2 , SRMR, RMSEA) and parameter estimates are validated. Taken together with the flexibility of SAS in data management and programming, PROC TCALIS provides a convenient alternative tool for researchers to conduct both applied and simulated invariance studies.

Acknowledgements

We would like to thank Marilyn Thompson and Samuel Green for granting permission to use the example data. This work benefited from comments by Yanyun Yang, and is supported by the University of Kansas Life Span Institute/Research Design & Analysis unit (LSI/RDA) and the Center for Educational Testing & Evaluation (CETE).

References

- Arbuckle, J. L. (2003). *AMOS (Version 5.0)* [Computer software]. Chicago: Smallwaters.
- Bentler, P. M., & Wu, E. J. C. (2002). *EQS 6 for Windows guide*. Encino, CA: Multivariate Software.
- Bollen, K. A. (1989). *Structural equations with latent variables*. New York: Wiley.
- Brown, T. A. (2006). *Confirmatory factor analysis for applied research*. New York: Guilford Press.
- Browne, M. W. (1982). Covariance structures. In D. M. Hawkins (Ed.), *Topics in applied multivariate analysis* (pp. 72-141). Cambridge: Cambridge University Press.
- Browne, M. W. (1984). Asymptotically distribution-free methods for the analysis of covariance structures. *British Journal of Mathematical and Statistical Psychology*, 37, 62-83.
- Byrne, B. M. (2004). Testing for multigroup invariance using AMOS graphics: A road less traveled. *Structural Equation Modeling*, 11(2), 272-300.
- Byrne, B. M., Shavelson, R. J., & Muthén, B. (1989). Testing for the equivalence of factor covariance and mean structures: The issue of partial measurement invariance. *Psychological Bulletin*, 105, 456-466.
- Chen, F. F. (2007). Sensitivity of goodness of fit indexes to lack of measurement invariance. *Structural Equation Modeling*, 14(3), 464-504.

- Cheung, G. W., & Rensvold, R. B. (2002). Evaluating goodness-of-fit indexes for testing measurement invariance. *Structural Equation Modeling*, 9(2), 233–255.
- Chou, C.-P., & Bentler, P. M. (1990). Model modification in covariance structure modeling: A comparison among likelihood ratio, Lagrange multiplier, and Wald tests. *Multivariate Behavioral Research*, 25, 115–136.
- Fan, X., & Fan, X. (2005). Using SAS for Monte Carlo simulation research in SEM. *Structural Equation Modeling*, 12(2), 299–333.
- Fan, X., Felsövályi, Á., Sivo, S. A., & Keenan, S. (2003). *SAS for Monte Carlo studies: A guide for quantitative researchers*. Cary, NC: SAS Institute.
- Fan, X., & Sivo, S. A. (2005). Sensitivity of fit indexes to misspecified structural or measurement model components: Rationale of two-index strategy revisited. *Structural Equation Modeling*, 12(3), 343–367.
- Fan, X., & Sivo, S. A. (2009). Using Δ goodness-of-fit indexes in assessing mean structure invariance. *Structural Equation Modeling*, 16, 54–69.
- Jones-Farmer, L. A., Pitts J. P., & Rainer, R. K. (2008). A note on multigroup comparisons using SAS PROC CALIS. *Structural Equation Modeling*, 15, 154–173.
- Jöreskog, K. G., & Sörbom, D. (1996). *LISREL 8: User's reference guide*. Chicago: Scientific Software International.
- Little, T. D. (1997). Mean and covariance structures (MACS) analyses of cross-cultural data: Practical and theoretical issues. *Multivariate Behavioral Research*, 32, 53–76.
- Lubke, G. H., & Muthén, B. O. (2004). Applying multi-group confirmatory factor models for continuous outcomes to Likert scale data complicates meaningful group comparisons. *Structural Equation Modeling*, 11, 514–534.
- Marcoulides, G. A., & Hershberger, S. L. (1997). *Multivariate statistical methods: A first course*. Mahwah, NJ: Erlbaum.
- Marsh, H. W., Muthén, B. O., Asparouhov, T., Lüdtke, O., Robitzsch, A., Morin, A. J. S., & Trautwein U. (2009). Exploratory structural equation modeling, integrating CFA and EFA: Application to students' evaluations of university teaching. *Structural Equation Modeling*, 16(3), 439–476.
- McDonald, R. P., & Ho, M.-H. R. (2002). Principles and practice in reporting structural equation analyses. *Psychological Methods*, 7(1), 64–82.
- Meredith, W. (1993). Measurement invariance, factor analysis and factorial invariance. *Psychometrika*, 58, 525–543.
- Millsap, R. E., & Yun-Tein, J. (2004). Assessing factorial invariance in ordered-categorical measures. *Multivariate Behavioral Research*, 39(3), 479–515.
- Muthén, L. K., & Muthén, B. O. (1998–2007). *Mplus user's guide*. Los Angeles: Muthén & Muthén.
- Neale, M. C., Boker, S. M., Xie, G., & Maes, H. H. (2003). *Mx: Statistical modeling* (6th ed.). Richmond, VA: Department of Psychiatry, Virginia Institute for Psychiatric and Behavioral Genetics, Virginia Commonwealth University.
- Saris, W., Satorra, A., & Van Der Veld, W. (2009). Testing structural equation models or detection of misspecifications? *Structural Equation Modeling*, 16(4), 561–582.
- SAS Institute (2008). *SAS/STAT[®] 9.2 User's Guide*. Cary, NC: SAS Institute Inc.
- Steiger, J. H. (1998). A note on multiple sample extensions of the RMSEA fit index. *Structural Equation Modeling*, 5(4), 411–419.
- Thompson, M. S., & Green, S. B. (2006). Evaluating between-group differences in latent variable means. In G. R. Hancock & R. O. Mueller (Eds.), *Structural equation modeling: A second course* (pp. 119–169). Greenwich, CT: Information Age Publishing.
- Vandenberg, R. J., & Lance, C. E. (2000). A review and synthesis of the measurement invariance literature: Suggestions, practices, and recommendations for organizational research. *Organizational Research Methods*, 3, 4–69.
- Yang, Y., & Green, S. B. (2010). A note on structural equation modeling estimates of reliability. *Structural Equation Modeling*, 17(1), 66–81.